
文件中的字体规则 
 
此手册中出现的字体规则。 
 

字体规则               描述 
 
[表示内容]                    双引号内的资料长度最大 2*1024 bytes; 
 
 
<ESC> 脱离符号 (ASCII 27)，为一立即指令，

不管打印机是否为待机或忙碌中，均立即

回复打印机目前将态。 
 
 
~      (ASCII 126)，当打印机于待机状态时可

回传打印机状态。 
 
 
 
 
注：200 DPI: 1 mm = 12 dots 粗斜体字型表示批注。 
 
 
 
DOWNLOAD “TEST.BAS”         当所列出的内容为程序时以Arial五号字 
SET COUNTER @1 1              型表示。 
@1=”0001” 
TEXT 10,10,”3”,0,1,1,@1 
PRINT 3,2 
EOP

系统设定指令 
 
 SIZE 
 
功能：定义标签纸的宽度及高度 
 
语法：(1) 英制系统 (英寸) 
          SIZE m,n 
      (2) 公制系统 (厘米) 
          SIZE m mm, n mm 
 

参   数      

 

说       明 
m    定义标签纸的宽度 (英寸或厘米) 
n    定义标签纸的宽度 (英寸或厘米) 

 
注意事项:  

200 DPI: 1 mm = 8 dots 
300 DPI: 1mm = 12 dots 
在公制系统下, 在参数 m 与 “mm” 之间必须要有空格分隔 

 
 

范例： 
(1) 英制系统 (英寸) 

SIZE 3.5, 3.00 
(2) 公制系统 (厘米) 

SIZE 100 mm, 100 mm 
 

 
 

其他参考项目： 
GAP, BLINE 

 GAP  
 
功能：定义两张标签之间的间隙距离 
 
语法：(1) 英制系统 (英寸) 
          GAP m,n 
      (2) 公制系统 (厘米) 
          GAP m mm, n mm 

 
参   数      

注意事项: 
在公制系统下, 在参数 m 与 “mm” 之间必须要有空格分隔 
另当标签纸由 “黑标纸张” 换成 “间隙标签” 时, 请先下达 “GAP” 的设 
定指令 
如：在 DOS 模式下 
    C:\>COPY CON LPT1 <Enter> 
        GAP 2 mm,0<Enter> 
        <Ctrl>+<Z><Enter> 

 
 
 
 
 

说          明 
m    定义标签间隙高度 (英寸或厘米) 
                 0 ≤ m ≤ 1 (inch), 0 ≤ m ≤ 25.4 (mm) 
n    定义标签间隙高度的补偿值 (英寸或厘米) 

[ ]n ≤ label length (inch or mm) 
0,0              表示连续纸模式 

 

范例： 
一般标签纸 
(1) 英制系统 (inch) 

GAP 0.12,0 
(2) 公制系统 (mm) 

GAP 3 mm,0 
(3) 连续纸模式 

GAP 0,0 
 

 
 
 
 

特殊标签纸 

(1) 英制系统 (inch) 
GAP 0.30, 0.10 

(2) 公制系统 (mm) 
GAP 7.62 mm, 2.54 mm 
 

 
 
 

其他参考项目： 
SIZE, BLINE 

 BLINE 
 
功能：设定黑标高度及使用者定义标签印完后标签额外送出的长度 
 
语法：(1) 英制系统 (英寸) 

            BLINE m,n 
        (2) 公制系统 (厘米) 
             BLINE m mm, n mm 

 
 参   数        

范例： 

说          明 
 m     黑标高度 
      0.1 ≤ m ≤ 1 (inch), 2.54 ≤ m ≤ 25.4 (mm) 
 n     额外送出纸张长度  0 ≤ n ≤ label length 
0,0                    连续纸 
 
注意事项： 
在公制系统下, 在参数 m 与 “mm” 之间必须要有空格分隔。另当标签纸由“间隙标签” 
换成 “黑标纸张” 时, 请先下达 “BLINE” 的设定指令 
如：在 DOS 模式下 
    C:\>COPY CON LPT1 <Enter> 
        BLINE 2 mm,0<Enter> 
        <Ctrl>+<Z><Enter> 

(1) 英制系统 (英寸) 
BLINE 0.20,0.50 
 

(2) 公制系统 (厘米) 
BLINE 5.08 mm,12.7 mm 
 

 
其他参考项目： 

SIZE, GAP 

 OFFSET 
 
功能：定义标签于打印完后额外推出的长度，特別是在使用自动剥纸器或裁刀功能时，可用来

调校标签停止的位置，在打印下一张时打印机会将原先多推出或少推出的部分以回拉方

式补偿回来。此方式适用于剥离模式和裁刀模式。 
    

语法：(1) 英制系统 (英寸) 
     OFFSET m 

      (2) 公制系统 (厘米) 
             OFFSET m mm 

 
参数           

范例： 

说          明 
m     偏移距离 (英寸或厘米) 
      -1≤ m ≤1(inch) 
       
警告：不适当的偏移值会导致出现 “卡纸” 的现象 
 

(1) 英制系统 (英寸) 
OFFSET 0.5 
 

(2) 公制系统 (厘米) 
OFFSET 12.7 mm 
 
 

 
其他参考项目： 

SIZE, GAP, SET PEEL 

 SPEED  
 
功能：设定打印机的打印速度 
 
语法：SPEED n 

 
参     数   

N 

说          明 
n    每秒的打印速度，以英寸计算 
 

2 3 4 5 6 
9034T o o o   
9035T o o o o  

 
 范例： 

SPEED 4 
 

其他参考项目： 
DENSITY 

 DENSITY 
 

功能：设定打印机的打印浓度 
 
语法：DENSITY n 

 
参     数   

    范例： 

说          明 
n     0~15 

0 表示最淡的浓度 
      15 表示最深的浓度 

 

DENSITY 7 
 

其他参考项目： 
SPEED 

 DIRECTION 
 
功能：设定打印方向，此设定将被记录于 EEPROM 

    
语法：DIRECTION n 

 
参     数   

 

说          明 
n     0 或 1 请参考下方图例 
 

 
 

 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
  

范例： 
DIRECTION 0[,0] 
 

其他参考项目：： 
 
  



 

9 
THERMAL BARCODE PRINTER SERIES 

 

 REFERENCE 
 
功能：定义标签纸上的相對于原点的参考点座标，请参考下方图例 

 

 
 
 语法： 

REFERENCE x, y 
 
参       数   

注意事项：200 DPI: 1 mm = 8 dots 

说          明 
x     水平座标，单位 “dot”  
y     垂直座标，单位 “dot” 
 

   300 DPI: 1 mm = 12 dots 
 
 范例： 

REFERENCE 10,10 
 

其他参考项目： 
DIRECTION 

 CODEPAGE 
 

功能：该指令用于选择对应的国际字符集 
 
语法：CODEPAGE n 

 
参     数            说          明 
n                8-bit code page 字符集代码号 

437: United States 
850: Multilingual 
852: Slavic 
860: Portuguese 
863: Canadian/French 
865: Nordic 
 

 范例： 

Windows code page 
1250: Central Europe 
1252: Latin I 
1253: Greek 
1254: Turkish 

 

CODEPAGE 437 

 CLS 
 
功能：清除数据缓存  

    
语法：CLS 

 
注意事项：此项指令必须置于 SIZE 指令之后 

 
范例： 

CLS 
 

其他参考项目： 
SIZE, GAP, BLINE 

 FEED 
 

功能：将标签纸向前推送指定的长度 
 
语法：FEED n 

 
参     数   

 范例： 

说          明 
n     单位：dot 
      1 ≤ n ≤ 9999 
 

FEED 40 
 
注意事项：200 DPI: 1 mm = 8 dots 

     300 DPI: 1 mm = 12 dots 
 
其他参考项目： 

BACKFEED, SIZE, GAP, BLINE, HOME, FORMFEED 

 BACKFEED&BACKUP 
 

功能：将标签纸向后回拉指定的长度 
    

语法：BACKFEED n 
 
参     数   

注意事项：200 DPI: 1 mm = 8 dots 

说          明 
n     单位：dot 
      1 ≤ n ≤ 9999 
 
警告：不当的回拉距离会造成 “卡纸” 或 “碳带皱折” 的现象 

    300 DPI: 1 mm = 12 dots 
 

 范例： 
BACKFEED 40 
BACKUP 40 

 
其他参考项目： 

FEED, SIZE, GAP, BLINE, HOME, FORMFEED 

 FORMFEED 
 
功能：将标签向前推送至下一张标签的起始位置 
 
语法：FORMFEED 

 
范例： 

SIZE 4,2.5 
GAP 0 mm,0 
SPEED 4 
DENSITY 7 
DIRECTION 0 
OFFSET 0.00 
REFERENCE 0,0 
SET PEEL OFF 
SET COUNTER @0 +1 
@0="000001" 
FORMFEED 
CLS 
BOX 1,1,360,65,12 
TEXT 25,25,"3",0,1,1,"FORMFEED COMMAND TEST" 
TEXT 25,80,"3",0,1,1,@0 
PRINT 3,1 
 

其他参考项目： 
FEED, SIZE, GAP, BLINE, HOME, BACKFEED 

 HOME  
 

功能：在使用含有间隙或黑标的标签纸时，若不能确定第一张标签纸是否在正确打印位置时，

此指令可将标签纸向前推送至下一张标签纸的起点开始打印。 
 
语法：HOME 

 
范例： 

SIZE 4,2.5 
GAP 2 mm,0 
SPEED 4 
DENSITY 7 
DIRECTION 0 
OFFSET 0.00 
REFERENCE 0,0 
SET PEEL OFF 
SET COUNTER @0 +1 
@0="000001" 
HOME 
CLS 
BOX 1,1,360,65,12 
TEXT 25,25,"3",0,1,1,"HOME COMMAND TEST" 
TEXT 25,80,"3",0,1,1,@0 
PRINT 3,1 

 
其他参考项目： 

FEED, SIZE, GAP, BLINE, FORMFEED 

 PRINT 
 
功能：将存于数据缓存的标签打印 

    
语法：PRINT m [,n] 

 
参     数   

 范例： 

说          明 
m    打印张数 
      1 ≤ m ≤ 999999999 
 
N                 每张标签需重复打印的张数 
      1 ≤ n ≤ 999999999 
  

SIZE 60 mm, 20 mm 
SET COUNTER @1 1 
@1=”0001” 
CLS 
TEXT 10,10,”3”,0,1,1,@1 
PRINT 3,2 
 

 
 
 
 
 
 

        
其他参考项目： 

SET COUNTER, INPUT, DOWNLOAD 

 SOUND 
 

功能：控制打印机蜂鳴器发聲的頻率，共有 10 阶，并可由 “interval” 参数设定音与音的间隔

时间 
    

语法：SOUND level,interval 
 
参     数   说         明 
level    音阶：0~9 
interval   间隔时间: 1~4095 

 
范例： 

SOUND 5,200 
SOUND 3,200 
SOUND 3,200 
SOUND 4,200 
SOUND 2,200 
SOUND 2,200 
SOUND 1,200 
SOUND 2,200 
SOUND 3,200 
SOUND 4,200 
SOUND 5,200 

 LIMITFEED 
 

功能：限定定间隙校正执行的最大长度，若在此长度范围内无法测得间隙存在，则将感应器模

式定在连续纸模式下 
 

   语法：(1) 英制系统 (英寸) 
LIMITFEED n  

 (2) 公制系统 (厘米) 
LIMITFEED n mm 

 
参     数   说          明 
n       间隙校正的最大长度 

 
范例： 
     LIMITFEED 10 
 
注： 

1. 此项设定会被记录于 EEPROM 中 

2. 初始化的预设最大长度为 4英寸 

3. 在公制系统中，参数 m 与单位 “mm” 之间必须有空格分隔 

 SELFTEST 
 

功能：不经自测动作，直接将打印机讯息印在标签纸上。 
    

语法： 
SELFTEST 

 
   范例： 

SELFTEST 

卷标内容设计指令 
 
 BAR 
 

功能：画线或长条物 
    

语法：BAR x, y, width, height 
 
参     数   

 

说     明 
x    线条左上角 X 座标，单位 dot 
y    线条左上角 Y 座标，单位 dot 
width    线条宽度，单位 dot 
height   线条高度，单位 dot 
 

注意事项：200 DPI: 1 mm = 8 dots 
      300 DPI: 1 mm = 12 dots 

 
 范例： 

SIZE 4,2.5 
GAP 0,0 
SPEED 6 
DENSITY 8 
DIRECTION 0 
CLS 
BAR 100, 100, 300, 200 
PRINT 1,1 

 
 

其他参考项目： 
BOX 

 BARCODE 
 

功能：印制一维条码，下方为所支持的条码列表。 
 
 Code 128 (自动转换字符子集) 
 Code 128M (手动转换字符子集) 
 EAN 128 (自动转换字符子集) 
 Interleaved 2 of 5 
 Interleaved 2 of 5 with check digit 
 Code 39 
 Code 39 with check digit 
 Code 93 
 EAN 13 
 EAN 13 with 2 digits add-on 
 EAN 13 with 5 digits add-on 
 EAN 8 
 EAN 8 with 2 digits add-on 
 EAN 8 with 5 digits add-on 
 Codabar 
 Postnet 
 UPC-A 
 UPC-A with 2 digits add-on 
 UPC-A with 5 digits add-on 
 UPC-E 
 UPC-E with 2 digits add-on 
 UPC-E with 5 digits add-on 
 MSI 
 PLESSEY  
 China POST(矩阵 25 码) 
 ITF14  
 EAN14 

语法： 
BARCODE X, Y, ”code type”, height, human readable, rotation, narrow, wide, “code” 
 
参     数   

Control code 

说          明 
X     明定条码左上角 X 座标 
Y     明定条码左上角 X 座标 
条码類型    

128    Code 128，自动转换 A、B、C 的条码格式。  
128M   Code 128，手动转换 A、B、C 的条码格式。 

 
A B C 

096 FNC3 FNC3 NONE 
097 FNC2 FNC2 NONE 
098 SHIFT SHIFT NONE 
099 CODE C CODE C NONE 
100 CODE B FNC4 CODE B 
101 FNC4 CODE A CODE A 
102 FNC1 FNC1 FNC1 
103 Start (CODE A) 
104 Start (CODE B) 
105 Start (CODE C) 

 
“!”是做条码格式转换使用，若未给定起始条码的格式时，则以 code B 做为起始条码。 
 
EAN128   Code 128, 自动转换 A，B，C 条码格式 

   25    Interleaved 2 of 5 
   25C    Interleaved 2 of 5 含校验码 
   39    Code 39 
   39C    Code 39 含校验码 
   93    Code 93  
   EAN13   EAN 13 
   EAN13+2  EAN 13 with 2 digits add-on 
   EAN13+5  EAN 13 with 5 digits add-on 
   EAN8   EAN 8  
   EAN8+2   EAN 8 with 2 digits add-on 
   EAN8+5   EAN 8 with 5 digits add-on 
   CODA   Codabar 
   POST   Postnet 
   UPCA   UPC-A 
   UPCA+2   UPC-A with 2 digits add-on 
   UPCA+5   UPC-A with 5 digits add-on 
   UPCE   UPC-E  
   UPCE+2   UPC-E with 2 digits add-on 
   UPCE+5   UPC-E with 5 digits add-on 
   CPOST   China post code 
   MSI    MSI code 
   MSIC   MSI code with check digital 
   PLESSEY  PLESSEY code 
   ITF14   ITF 14 code 
   EAN14   EAN 14 code 
 

height    条码高度 (dot) 
human readable  0 : 无人眼可识别码 
      1 : 有人眼可识别码 

rotation    顺时针旋转条码角度 
   0    不旋转 
   90    顺时针旋转 90 度 
   180    顺时针旋转 180 度 
   270    顺时针旋转 270 度  

narrow    窄条码比例因子 (dot) 
wide     宽条码比例因子 (dot) 
 

 narrow : wide 
1:1 

narrow : wide 
1:2 

narrow : wide 
1:3 

narrow : wide 
2:5 

narrow : wide 
3:7 

128 10x - - - - 
EAN128 10x - - - - 

25 - 10x 10x 5x - 
25C - 10x 10x 5x - 
39 - 10x 10x 5x - 

39C - 10x 10x 5x - 
93 - - 10x - - 

EAN13 8x - - - - 
EAN13+2 8x - - - - 
EAN13+5 8x - - - - 

EAN 8 8x - - - - 
EAN 8+2 8x - - - - 
EAN 8+5 8x - - - - 
CODA - 10x 10x 5x - 
POST 1x - - - - 
UPCA 8x - - - - 

UPCA+2 8x - - - - 
UPCA+5 8x - - - - 

UPCE 8x - - - - 
UPCE+2 8x - - - - 
UPCE+5 8x - - - - 
CPOST - - - - 1x 

MSI - - 10x - - 
MSIC   10x  - 

PLESSY - - 10x - - 
ITF14 - 10x 10x 5x - 

EAN14 - - - 5x - 
 

Barcode type Maximum bar code length 
128 - 

EAN128 - 
25 - 

25C - 
39 - 

39C - 
93 - 

EAN13 12 
EAN13+2 14 
EAN13+5 17 

EAN 8 7 
EAN 8+2 9 
EAN 8+5 12 

CODA - 
POST 5,9,11 
UPCA 11 

UPCA+2 13 
UPCA+5 16 

UPCE 6 
UPCE+2 8 
UPCE+5 11 
CPOST - 

MSI - 
MSIC  

PLESSY - 
ITF14 13 
EAN14 13 

  
范例： 

BARCODE 100,100,”39”,96,1,0,2,4,”1000” 
  BARCODE 10,10,”128M”,48,1,0,2,2,”!104!096ABCD!101EFGH” 
 
         (上述 CODE 128M 的范例为使用 CODE B 起始的条码，其中!096(FNC3)与 ABCD 均以

CODE B 方式编码；!101 为将原编码方式由 CODE B 转换为 CODE A，后续的 EFGH
即为使用 CODE A 方式编码)  

 BITMAP 
 

功能：绘制 BITMAP 格式的图形 (非 BMP 格式的图档) 
    

语法：BITMAP X, Y, width, height, mode, bitmap data… 
 
参     数   

范例： 

说          明 
X    明定图形左上角 X 座标 
Y    明定图形左上角 Y 座标 
width    图形的宽度，单位为 byte 
height   图形的高度，单位为 dot 
mode    绘制图形的方式 
 0    OVERWRITE 
 1    OR 
 2        XOR 
bitmap data  图形资料  

 

 
 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

ROW 
(Y- axis) 

L-Byte R-Byte 
Binary Hexadecimal Binary Hexadecimal 

0 00000000 00 00000000 00 
1 00000000 00 00000000 00 
2 00000000 00 00000000 00 
3 00000111 07 11111111 FF 

4 00000011 03 11111111 FF 
5 00010001 11 11111111 FF 
6 00011000 18 11111111 FF 
7 00011100 1C 01111111 7F 
8 00011110 1E 00111111 3F 
9 00011111 1F 00011111 1F 
A 00011111 1F 10001111 8F 
B 00011111 1F 11000111 C7 
C 00011111 1F 11100011 E3 
D 00011111 1F 11110111 F7 
E 00011111 1F 11111111 FF 
F 00011111 1F 11111111 FF 

 
范例: 
SIZE 4,2 
GAP 0,0 
CLS 
BITMAP 200,200,2,16,0,                                                             �-?????  
PRINT 1,1 

 
Hexadecimal ASCII 

53 49 5A 45 20 34 2C 32 0D 0A 47 41 50 20 
30 2C 30 0D 0A 43 4C 53 0D 0A 42 49 54 4D 
41 50 20 32 30 30 2C 32 30 30 2C 32 2C 31 
36 2C 30 2C 00 00 00 00 00 00 07 FF 03 FF 
11 FF 18 FF 1C 7F 1E 3F 1F 1F 1F 8F 1F C7 
1F E3 1F E7 1F FF 1F FF 0D 0A 50 52 49 4E 
54 20 31 2C 31 0D 0A 

SIZE 4,2 
GAP 0,0 
CLS 
BITMAP 200,200,2,16,0,      
                                                       �-?-
????  
PRINT 1,1 
 

 
 

其他参考项目： 
PUTBMP, PUTPCX 

 BOX 
 

功能：绘制方框 
 
语法：BOX X_start, Y_start, X_end, Y_end, line thickness 

 
参     数   

范例： 

说          明 
X_start   方框左上角 X 座标，单位：dot 
Y_start   方框左上角 Y 座标，单位：dot 
X_end   方框右下角 X 座标，单位：dot 
Y_end   方框右下角 X 座标，单位：dot 
line thickness  方框线条粗細，单位：dot 

 
注意事项：200 DPI: 1 mm = 8 dots 

  300 DPI: 1 mm = 12 dots 
 

SIZE 4,2.5 
GAP 0,0 
SPEED 6 
DENSITY 8 
DIRECTION 0 
CLS 
BOX 100,100,200,200,5 
PRINT 1,1 
 
 

其他参考项目： 
BAR 

 ERASE 
 
功能：清除影像缓冲区中指定大小的区域 

    
语法：ERASE X_start, Y_start, X_width, Y_height 

 
参     数   

 

说          明 
X_start   欲清除区域的左上角 X 座标，单位：dot 
Y_start   欲清除区域的左上角 Y 座标，单位：dot 
X_width   欲清除区域的宽度，单位：dot 
Y_height   欲清除区域的高度，单位：dot 

范例： 
SIZE 4,2.5 
GAP 0,0 
SPEED 6 
DENSITY 8 
DIRECTION 0 
CLS 
BAR 100, 100, 300, 300 
ERASE 150,150,200,200 
PRINT 1,1 
 
 

其他参考项目： 
CLS 

 PUTBMP 
 

功能：打印 BMP 格式图档 
    

语法：PUTBMP X, Y, ”filename” 
 
参     数   

 范例：^Z 表示<Ctrl>+<z>或者是<F6>键 

说          明 
X    BMP 图形左上角 X 座标 
Y    BMP 图形左上角 Y 座标 
filename   已载至打印机的 BMP 图档 
 

注意事项：建议使用仅有黑白二种颜色的 BMP 图档 
 

   
   
   
   
   
   
   
   
   
   
   
   
   
   
   
   
   
   
 
 
 
 
 
 
 
   

其他参考项目： 
DOWNLOAD, BITMAP, PUTPCX 

  



 

30 
THERMAL BARCODE PRINTER SERIES 

 

 PUTPCX 
 

功能：打印 PCX 格式图档 
   语法：PUTPCX X, Y, ”filename” 

 
参     数   

 

说          明 
X     PCX 图形左上角 X 座标 
Y     PCX 图形左上角 Y 座标 
filename    已载至打印机的 PCX 图档 
 

范例：^Z 表示<Ctrl>+<z>或者是<F6>键 
   
         
         
         
         
         
         
   
   
   
   
   
   
   
   
   
   
   
   
         
   
   

 
 
其他参考项目： 

DOWNLOAD, BITMAP, PUTPCX 

 QRCODE 
 

功能：绘制 QRCODE 二维条码 
   

语法： 
QRCODE X, Y, ECC Level, cell width, mode, rotation, [model, mask,]"Data string” 

参     数   

4).日文汉字字符 (Shift JIS values 8140HEX –9FFCHEX and E040HEX –EAA4 HEX. 
These are values shifted from those of JIS X 0208. Refer to JIS X 0208 Annex 1 
Shift Coded Representation for detail.). 

  
条码的最大资料长度： 
                     

说          明 
X    QRCODE 条码左上角 X 座标 
Y    QRCODE 条码左上角 Y 座标 
ECC level   错误纠正能力等级 

L    7% 
  M    15% 
  Q    25% 
  H    30% 
cell width   1~10 
mode    自动生成编码/手动生成编码 

A    Auto 
  M    Manual 
rotation   顺时针旋转角度 

0    不旋转 
  90    顺时针旋转 90 度 
  180    顺时针旋转 180 度 
  270    顺时针旋转 270 度 
model   条码生成样式 

1     (预设), 原始版本 
2    扩大版本 

mask    范围：0~8，预设 7 
Data string   条码资料内容 
 
可用编码字符集： 

1). 数值资料：数字 0~9 
2). 文字数字资料：数字 0~9；大写字母 A-Z; 其他：space，$%*+-./:  
3). 8-bit 二进制资料(JIS 8-bit 字符表(拉丁语及假名) 与 JIS X 0201 相符) 

Model 1 (Version 14-L)   

5. “!”则是用来转换资料的格式，”N”、”A”、”B”、”K”等资料型态可透过”!”

Model 2 (Version 40-L) 
          1). 数值资料：        1,167 characters        7,089 characters 

2). 文字数字资料：   707 characters          4,296 characters 
3). 8-bit 二进制资料：  486 characters          2,953 characters 
4). 日文汉字资料:      299 characters          1,817 characters 

 
于手动生成编码下： 

1. 若条码内容的第一个字符为”A”，则后续的资料为”文字数字”型态。 

2. 若条码内容的第一个字符为”N”，则后续的资料为”数字”型态。 

3. 若条码内容的第一个字符为”B”，则后续的 4 位数字表示紧接二进位资料的长度

(单位为 byte)，且后续的资料为”二进制资料”型态。 

4. 若条码内容的第一个字符为”K”，则后续的资料为”日文汉字”型态。 

的转换组成一组条码内容。 
    
 范例：   手动生成 QRCODE 的范例: 

QRCODE 100,10,L,7,M,0,1,1,"ATHE FIRMWARE HAS BEEN UPDATED" 
(A：文字数字混合型态的资料)  
QRCODE 100,10,M,7,M,0,1,2,"N123456" 
(N：数值型态的资料)  
QRCODE 100,10,Q,7,M,0,1,3,"N123456!ATHE FIRMWARE HAS BEEN 
UPDATED" 
(N：数值型态的资料；!：资料型态转换用字符；A；文字数字混合型态的资料) 
QRCODE 100,10,H,7,M,0,1,3,"B0012Product name" 
(B：二进制资料；0012：12 bytes ) 
QRCODE 100,10,M,7,M,0,M1,S3,"K 打印机" 
(K：日文汉字资料) 

 
自动生成 QRCODE 的范例: 
QRCODE 100,10,M,7,A,0,"THE FIRMWARE HAS BEEN UPDATED" 

 

(1) 自动生成 QRCODE 的范例： 
a. 一般资料字符串 

SIZE 4,2.5 
GAP 0.12,0 
CLS 
QRCODE 10,10,H,4,A,0,"ABCabc123" 
QRCODE 160,160,H,4,A,0,"123ABCabc" 
QRCODE 310,310,H,4,A,0,"打印机 ABCabc123" 

PRINT 1,1 
b. 资料字符串中包含<Enter>(即 0Dh, 0Ah) 

SIZE 4,2.5 
GAP 0.12,0 
CLS 
QRCODE 10,10,H,4,A,0,"ABC<Enter> 
abc<Enter> 
123" 
QRCODE 160,160,H,4,A,0,"123<Enter> 
ABC<Enter> 
abc" 
QRCODE 310,310,H,4,A,0,"打印机<Enter> 
ABC<Enter> 
abc<Enter> 
123" 

PRINT 1,1 

c. 资料串接 (必须在 DOENLOAD…EOP 模式下) 
DOWNLOAD “DEMO.BAS” 
SIZE 4,2.5 
CAP 0.12,0 
CLS 
QRCODE 10,10,H,4,A,0,"ABCabc123"+STR$(1234) 
QRCODE 160,160,H,4,A,0,"123ABCabc"+"1234" 
QRCODE 310,310,H,4,A,0,"打印机 ABCabc123"+"1234"+"abcd" 
PRINT 1,1 
EOP 
DEMO 

d. 资料内容含有双引号(“)时，请用\[“]取代 
SIZE 4,2.5 
CAP 0.12,0 
CLS 
QRCODE 10,10,H,4,A,0,"ABC\["]abc\["]123" 
QRCODE 160,160,H,4,A,0,"123\["]ABC\["]abc" 
QRCODE 310,310,H,4,A,0,"\["]打印机\["]ABCabc123" 
PRINT 1,1 

(2) 手动生成 QRCODE 的范例： 
a. 一般资料字符串： 

SIZE 4,2.5 
CAP 0.12,0 
CLS 
QRCODE 10,10,H,4,M,0,"AABC!B0003abc!N123" 
QRCODE 160,160,H,4,M,0,"N123!AABC!B0003abc" 
QRCODE 310,310,H,4,M,0,"K 打印机!AABC!B0006abc123" 
PRINT 1,1 

b. 资料字符串中包含<Enter>(即 0Dh, 0Ah), <Enter> 则为 8-bit byte data 
SIZE 4,2.5 
CAP 0.12,0 
CLS 
QRCODE 10,10,H,4,M,0,"AABC!B0007<Enter> 

abc<Enter> 
!N123" 
QRCODE 160,160,H,4,M,0,"N123!B0002<Enter> 
!AABC!B0005<Enter> 
abc" 
QRCODE 310,310,H,4,M,0,"K 打印机!B0002<Enter> 
!AABC!B0010<Enter> 
abc<Enter> 
123" 

PRINT 1,1 

c. 资料串接 (必须在 DOENLOAD…EOP 模式下) 
DOWNLOAD “A.BAS” 
SIZE 4,2.5 
CAP 0.12,0 
CLS 

QRCODE 10,10,H,4,M,0,"AABC!B0006abc123!N"+STR$(1234) 
QRCODE 160,160,H,4,M,0,"N123!AABC!B0007abc"+"1234" 
QRCODE 310,310,H,4,M,0,"K 打印机!AABC!B0014abc123"+"1234"+"abcd" 

      PRINT 1,1 
      EOP 
      A 

d. 资料内容含有双引号(“)时，请用\[“]取代 
SIZE 4,2.5 
CAP 0.12,0 
CLS 

QRCODE 10,10,H,4,M,0,"AABC!B0005\["]abc\["]!N123" 
QRCODE 160,160,H,4,M,0,"N123!B0001\["]!AABC!B0004\["]abc" 
QRCODE 310,310,H,4,M,0,"B0001\["]!K 打印机!B0010\["]ABCabc123" 

PRINT 1,1 

 REVERSE 
 

功能：将图像缓存中指定的区域做反相的动作 
    

语法：REVERSE X_start, Y_start, X_width, Y_height 
 
参     数   

 范例： 

说          明 
X_start   欲清除区域的左上角 X 座标，单位：dot 
Y_start   欲清除区域的左上角 Y 座标，单位：dot 
X_width   欲清除区域的宽度，单位：dot 
Y_height   欲清除区域的高度，单位：dot 
 
注意事项：200 DPI: 1 mm = 8 dots 

300 DPI: 1 mm = 12 dots 
 

SIZE 4,2.5 
GAP 0,0 
SPEED 6 
DENSITY 8 
DIRECTION 0 
CLS 
TEXT 100,100,"3",0,1,1,"REVERSE" 
REVERSE 90,90,128,40 
PRINT 1,1 

 TEXT 
 

功能：打印文字 
    

语法：TEXT X, Y, ”font”, rotation, x-multiplication, y-multiplication, “content” 
 
参     数   

7 14 x 25 英数字体 OCR-A 

说          明 
X    文字方塊左上角 X 座标 
Y    文字方塊左上角 Y 座标 
font    字型名称 
  1     8 x 12 英数字体 
  2     12 x 20 英数字体 
  3     16 x 24 英数字体 
  4     24 x 32 英数字体 
  5     32 x 48 英数字体 
  6     14 x 19 英数字体 OCR-B 

8 21 x 27 英数字体 OCR-B 
  TST24.BF2   繁体中文 24x24 字体(大五码) 
  TSS24.BF2   简体中文 24x24 字体(GB 码) 
  K     韩文 24x24 字体(KS 码) 
 

Rotation    顺时针旋转角度  
  0     不旋转 
  90     顺时针旋转 90 度 
  180     顺时针旋转 180 度 

270 顺时针旋转 270 度 
 

X-multiplication:  水平放大值，最大可放大至 10 倍  
有效系数：1~10 

Y-multiplication:  垂直放大值，最大可放大至 10 倍 
有效系数：1~10 
 

注意事项：若文字内容包括双引号(“)时，则请用\[“] 取代 
   若要打印 0D(hex)字符时，请在程序中使用\[R]来打印 CR 
   若要打印 0A(hex)字符时，请在程序中使用\[A]来打印 LF 
   五号字英文字母仅可打印大写字母 

 
 范例 1： 

SIZE 3,2 
GAP 0,0 
CLS 
TEXT 100,100,"5",0,1,1,"\["]DEMO FOR TEXT\["]" 
TEXT 100,200,"ROMAN.TTF",0,1,20,"\["]True Type Font Test Print\["]" 
PRINT 1,1 

 范例2： 
DOWNLOAD F,"test.BAS" 
SPEED 4 
DENSITY 8 
SET PEEL OFF 
DIRECTION 0 
SIZE 1.000,0.800 

GAP 0.12,0.00 
OFFSET 0.000 
REFERENCE 0,0 
:START 
CLS 
INPUT "Description",A$ 
Again: 
F$="" 
INPUT "Font:",1,F$ 
IF ASC(F$)>53 THEN GOTO Again 
IF ASC(F$)<49 THEN GOTO Again 
X_Again: 
INPUT "X_multiple:",1,XM 
IF XM>8 THEN GOTO X_Again 
IF XM<1 THEN GOTO X_Again 
Y_Again: 
INPUT "Y_multiple:",1,YM 
IF YM>8 THEN GOTO Y_Again 
IF YM<1 THEN GOTO Y_Again 
 
Z=LEN(A$) 
WWidth=Z*XM*8 
IF F$="1" THEN 
   REM *****When Font1, the space between every single word is XM*2 dots***** 
   WWidth=Z*(XM*8) 
   SWidth=Z*(XM*2) 
   X=101-((WWidth+SWidth)/2) 
ELSEIF F$="2" THEN 
   REM *****When Font2, the space between every single word is XM*2 dots***** 
   WWidth=Z*(XM*12) 
   SWidth=Z*(XM*2) 
   X=101-((WWidth+SWidth)/2) 
ELSEIF F$="3" THEN 
   REM *****When Font3, the space between every single word is XM-1 dots***** 
   WWidth=Z*(XM*16) 
   SWidth=Z*(XM-1) 
   X=101-((WWidth+SWidth)/2) 
ELSEIF F$="4" THEN 
   REM *****When Font4, the space between every single word is XM-1 dots***** 
   WWidth=Z*(XM*24) 
   SWidth=Z*(XM-1) 
   X=101-((WWidth+SWidth)/2) 
ELSEIF F$="5" THEN 
   REM *****When Font5, the space between every single word is XM*4 dots***** 
   WWidth=Z*(XM*32) 
   SWidth=Z*(XM*4) 
   X=101-((WWidth+SWidth)/2) 
ENDIF 
TEXT X,116,F$,0,XM,YM,A$ 
PRINT 1,1 
GOTO START 
EOP 

询问打印机状态指令 
 
 <ESC>!? 
 

 功能：透过串口送出指令用来获得打印机目前状态，其中<ESC>脱离符号表示 ASCII 27 (Hex1B)，
此项指令可随时被送出，即使是在打印机发生错误状态时。 
 
位      

语法：<ESC>!? 

状     态 
 0      就绪 
1         卡纸 
2      缺纸 
3      无碳带 
4      暂停打印 
5      打印中 
6      外盖开启 
7      过热 

 
 

 
其他参考项目： 

<ESC>!R 

 <ESC>!R 
 

功能：透过串口送出指令使打印机重新开机，其中<ESC>脱离符号表示 ASCII 27 (Hex1B)，此项

指令可随时被送出，重新开机后 DRAM 中的资料将被清除。 
 
语法：<ESC>!R 
 
其他参考项目： 

<ESC>!? 

 ~!@ 
 

功能：透过串口询問打印机已打印的里程数(单位：公尺)。里程数的整数部分将被回传，小数部

分则被忽略。回传的型态为 ASCII字符，结尾部分为 0x0D 0x0A 
    

语法：~!@ 
 

   范例： 
~!@ 

 ~!A 
 

功能：透过串口回传打印机内存剩余空间，回传时将分別列出 DRAM 和 FLASH 的剩余空间，

以十进制方式回传，并以 0x0D 0x0A 做结束。 
 

语法：~!A 
 
范例： 

~!A 
 
其他参考项目： 

FILES 

 ~!C 
 

功能：透过串口回传打印机是否有安装 RTC。 
 
语法：~!C 

 
回传值              

范例： 

说       明 
0                   未安装 RTC 
1                   已安装 RTC 
 

~!C 

 ~!D 
 

功能：进入“输入备份模式” 
    

语法：~!D 
    
   范例： 

~!D 

 ~!F 
 

功能：透过串口查询回传打印机内存在的文件名称或是字型名称。档名以 ASCII 字符回传，档

名与档名之间以 0x20 间隔，回传字符串的结束字符为 0x0D  0x1A。 
    

语法：~!F 
 
范例： 

~!F 
 

其他参考项目： 
FILES 

 ~!I 
 
功能：查询打印机内所设定的 CODE PAGE 值 
      经 RS-232 回传的格式为： 
      例：8 bit : 437,001 

         有关打印机回传信息，请参阅 CODEPAGE 指令 
    

语法：~!I 
 

范例： 
~!I 
 

其他参考项目：: 
CODEPAGE 

 ~!T 
 

功能：查询打印机的型号，经 RS-232 以 ASCII字符回传 
  

    
语法：~!T 

 
范例： 

~!T 
 

其他参考项目： 
~!I, ~!F 

文件管理指令 
 
 DOWNLOAD 

功能：”DOWNLOAD” 为一文件内容的首行，定义文件可被储存于打印机的 DRAM 中。下载至

打印机的文件可区分为两种：程序文件及资料文件 (包括字符型文件、PCX 图形文件和

BMP 图形文件等) 
 

 指令语法： 
1. 下载程序文件： 

DOWNLOAD [n,]“FILENAME.BAS” 
  参     数    

n 指定储存文件的位置 
说      明 

 不指定时: 文件被储存于 DRAM 
F：文件被储存于 FLASH 

FILENAME.BAS   储存于打印机中的文件名称 
 
注意事项： 

(1). 大小写将表示不同的文件名称 
(2). 程序文件的扩展名必须为”.BAS” 
(3). 若不指定储存的位置，则文件一律载至 DRAM，存于 DRAM 的文件会

因电源关闭而消失 
  

2. 下载资料文件： 
DOWNLOAD [n,]“FILENAME”, DATA SIZE, DATA CONTENT… 
 

  参     数    
n 指定储存文件的位置 

说      明 

 不指定时: 文件被储存于 DRAM. 
 F：文件被储存于 FLASH 

FILENAME  储存于打印机中的文件名称 
DATA SIZE 不含文件头的实际文件大小，以字节数计算 

 
注意事项： 

(1). 行与行之间的资料以 CR(0x0D)及 LF(0x0A)做分隔 
(2). 若不指定储存的位置，则文件一律载至 DRAM，存于 DRAM 的文件会

因电源关闭而消失 
 

 范例： 
        下方程序范例会将文件载至打印机的 DRAM 
  DOWNLOAD “EXAMPLE.BAS” 
  SIZE 4,4 
  GAP 0,0 
  DENSITY 8 
  SPEED 6 
  DIRECTION 0 
  REFERENCE 0,0 
  SET PEEL OFF 
  CLS 
  TEXT 100,100,”3”,0,1,1,”EXAMPLE PROGRAM” 
  PRINT 1 
  EOP 

        注意事项：当编写一个可被下载至打印机的程序时，”DOWNLOAD”必须置于文件的首
行，”EOP”必须置于程序的最后一行。 

                  执行程序的方式有二：一为调用主档名(不含 BAS)执行，另一为使用 RUN 指
令执行程序。 

 
范例: 
1. 调用主档名执行： 

C:\>COPY CON LPT1<ENTER> 
EXAMPLE<ENTER> 
<CTRL><Z> 
C:\> 
 

2. 使用 RUN 指令执行程序： 
C:\>COPY CON LPT1<ENTER> 
RUN “EXAMPLE.BAS”<ENTER> 
<CTRL><Z> 
C:\> 

 
下方为下载资料文件的范例. 
DOWNLOAD “DATA”,20,COMPUTER<Enter> 
2001<Enter> 
21<Enter> 

 
        注意事项：上方范例中<ENTER>表示按下键盘上的”ENTER”键。 

 
其他参考项目： 

EOP, RUN, PUTBMP, PUTPCX 

 EOP 
 

功能：作为载入程序文件的结尾。当使用 BASIC 语法时，DOWNLOAD “FILENAME”必须置于

文件首行，EOP 必须置于文件结尾。 
 

语法： 
EOP 
 

范例： 
DOWNLOAD “DEMO.BAS” 
SIZE 4,4  
GAP 0,0 
DENSITY 8 
SPEED 6 
DIRECTION 0 
REFERENCE 0,0 
SET PEEL OFF 
CLS 
TEXT 100,100,”3”,0,1,1,”DEMO PROGRAM” 
PRINT 1 
EOP 

 
其他参考项目： 

DOWNLOAD, EOP 

 FILES 
 

功能：此指令会将已载入打印机存储器的文件名称打打印来 
 

语法： 
FILES 
 

范例： 
     下列步骤为在 DOS 模式下透过并口或串口将打印机内的文件打打印来 

 
串口： 
C:\>MODE COM1 96,N,8,1<ENTER> 
C:\>COPY CON COM1<ENTER> 

FILES<ENTER> 
<CTRL><Z><ENTER> 

C:\> 
 
并口： 
C:\>COPY CON LPT1<ENTER> 

FILES<ENTER> 
<CTRL><Z><ENTER> 

 
注意事项：上方范例中<ENTER>表示按下键盘上的”ENTER”键，<CTRL>表示按下键盘

上的”Ctrl”键，<Z>表示按下键盘上的”Z”键。 
 

 
其他参考项目： 

~!F, KILL 

 KILL 
 

功能：删除存储器内的文件，万用字符”*”可将 DRAM 或 FLASH 内的所有资料删除。 
 

语法： 
KILL [n],”FILENAME” 
 

参     数    
n  指定要删除档除的位置 

说       明 

 未指定时：删除 DRAM 中的文件. 
F：删除主机板上 FLASH 中的文件 
 

注意事项：当未指定删除文件的位置时，将删除 DRAM 中的文件 
 
语法范例： 
1. KILL “FILENAME” 
2. KILL “*.PCX” 
3. KILL “*” 
4. KILL F,“FILENAME” 

 
 范例： 

    使用者可使用”SELFTEST”指令、”FILES”指令或”~!F”获得打印机存储器内的文件名称，

下列步骤可透过并行口删去指定位置的文件。 
 
C:\>COPY CON LPT1<ENTER> 

FILES<ENTER> 
<CTRL><Z><ENTER> 

C:\>COPY CON LPT1<ENTER> 
KILL “DEMO.BAS” <ENTER> 
<CTRL><Z><ENTER> 

C:\>COPY CON LPT1<ENTER> 
FILES<ENTER> 
<CTRL><Z><ENTER> 

 
注意事项：上方范例中<ENTER>表示按下键盘上的”ENTER”键，<CTRL>表示按下键盘

上的”Ctrl”键，<Z>表示按下键盘上的”Z”键。 
 

其他参考项目： 
~!F, FILES 

 MOVE 
 

功能：此指令可将原下载至 DRAM 的文件移至 FLASH 内 
 

语法： 
MOVE 
 

其他参考项目： 
DOWNLOAD, EOP 

 RUN 
 

功能：用来执行已存于存储器中的程序。 
 

语法： 
RUN “FILENAME.BAS” 
 

范例： 
C:\>COPY CON LPT1<ENTER> 
RUN “DEMO.BAS”<ENTER> 
<CTRL><Z><ENTER> 
C:\> 
 

注意事项：上方范例中<ENTER>表示按下键盘上的”ENTER”键，<CTRL>表示按下键盘
上的”Ctrl”键，<Z>表示按下键盘上的”Z”键。 

 
其他参考项目： 

DOWNLOAD, EOP 
  



 

53 
THERMAL BARCODE PRINTER SERIES 

 

BASIC指令及函数 
 
 ABS( ) 
 
    功能：此函数回传一整数或浮点数(小数)或一变量的绝对值。 

 
语法： 

ABS (-100) 
ABS (-99.99) 
ABS (VARIABLE) 
 

范例： 
DOWNLOAD “TEST.BAS” 
SIZE 4,4 
GAP 0,0 
DENSITY 8 
SPEED 3 
DIRECTION 0 
REFERENCE 0,0 
SET PEEL OFF 
CLS 
A=ABS(-100) 
B=ABS(-50.98) 
C=-99.99 
TEXT 100,100,”3”,0,1,1,STR$(A) 
TEXT 100,150,”3”,0,1,1,STR$(B) 
TEXT 100,200,”3”,0,1,1,STR$(ABS(C)) 
PRINT 1 
EOP 
 

其他参考项目： 
DOWNLOAD, EOP 

 ASC( ) 
 
  功能：回传一字符的 ASCII码。 

 
语法： 

ASC (“A”) 
 

范例： 
DOWNLOAD “TEST.BAS” 
SIZE 4,4 
GAP 0,0 
DENSITY 8 
SPEED 3 
DIRECTION 0 
REFERENCE 0,0 
SET PEEL OFF 
CLS 
CODE1=ASC(“A”) 
TEXT 100,100,”3”,0,1,1,STR$(CODE1) 
PRINT 1 
EOP 
 

其他参考项目： 
DOWNLOAD, EOP, STR$() 

 CHR$( ) 
 

功能：回传一指定 ASCII码所代表的字符。 
 

语法： 
CHR$(n) 
 
参     数   

 范例： 

说    明 
n    ASCII码. 

 

DOWNLOAD “TEST.BAS” 
SIZE 4,4 
GAP 0,0 
DENSITY 8 
SPEED 3 
DIRECTION 0 
REFERENCE 0,0 
SET PEEL OFF 
CLS 
A=65 
WORD$=CHR$(A) 
TEXT 100,100,”3”,0,1,1,WORD$ 
PRINT 1 
EOP 
 

其他参考项目： 
DOWNLOAD, EOP, STR$(), ASC$() 

 END 
 
    功能：主程序结束点 

 
 语法： 

END 
 

 范例： 
DOWNLOAD "DEMO.BAS" 
SIZE 4,2 
GAP 0,0 
DENSITY 8 
SPEED 6 
DIRECTION 0 
REFERENCE 0,0 
SET PEEL OFF 
CLS 
TEXT 200,60,"4",0,1,1,"END COMMAND TEST" 
X=300 
Y=200 
X1=500 
Y1=400 
GOSUB DR_LINE 
PRINT 1 
END 
 
:DR_LINE 
FOR I=1 TO 100 STEP 10 
BOX X+I,Y+I,X1-I,Y1-I,5 
NEXT 
RETURN 
EOP 
DEMO 
 

其他参考项目： 
DOWNLOAD, EOP 

 EOF( ) 
 

功能：当开启一已下载的文件时，判断文件是否已到结尾处。 
 

语法： 
EOF (File Handle) 
 
参     数   说          明 
File handle    0 或 1  
 
回   传   值   

范例： 

说          明 
非零     已到达文件尾端 
0     尚未到达文件尾端 
 

DOWNLOAD "DATA",16,COMPUTER 
2000 
 
DOWNLOAD "DEMO.BAS" 
SIZE 3,3 
GAP 0.0,0 
DENSITY 8 
SPEED 4 
DIRECTION 0 
REFERENCE 0,0 
SET PEEL OFF 
CLS 
OPEN "DATA",0 
SEEK 0,0 
Y=110 
TEXT 10,10,"3",0,1,1,"*****EOF TEST*****" 
:A 
Temp$="" 
READ 0,ITEM$,P 
TEXT 10,Y,"2",0,1,1,ITEM$+"$"+STR$(P)+"[EOF(0)="+STR$(EOF(0))+"]" 
BARCODE 10,Y+25,"39",40,1,0,2,4,"PRICE-"+STR$(P) 
Y=Y+100 
IF EOF(0)=0 THEN GOTO A 
PRINT 1 
EOP 
DEMO 
 

其他参考项目： 
DOWNLOAD, EOP, OPEN, READ, SEEK 

 OPEN 
 

功能：开启一已下载至打印机的文件，最多可同时开启 2 个文件。在开启文件之前必须先确定

文件已被下载至打印机的存储器内。 
 

语法： 
OPEN “Filename”, File handle 

 
参     数    

 

说          明 
Filename    已载至打印机的文件名称 
File handle    0 或 1  

范例： 
若有一文件名叫”DATA”的被载至打印机，且文件内容如下： 

 
DOWNLOAD "DATA1",56,COMPUTER 

2000 
12 
MOUSE 
500 
13 
KEYBOARD 
300 
100 

   
DOWNLOAD "DATA2",56,Computer 
3000 
32 
Mouse 
900 
93 
Keyboard 
700 
700 

 
        储存上述内容于文件名叫”DATA”的文件之中，下述步骤可将”DATA”内的资料下载至打印机

内。 
 
C:\>COPY DATA/B LPT1 
 
若有一个名叫”DEMO.BAS”的文件被下载，且其内容如下： 

 
DOWNLOAD "DEMO.BAS" 
SIZE 3,1 
GAP 0,0 
DENSITY 8 
SPEED 4 
DIRECTION 0 
REFERENCE 0,0 
SET PEEL OFF 
I=1 
Y=100 
GOSUB OpenData 
:Start 
CLS 

TEXT 10,10,"3",0,1,1,"*****OPEN COMMAND TEST*****" 
ITEM$="" 
READ 0,ITEM$,P,Q 
TEXT 10,Y,"2",0,1,1,ITEM$+"$"+STR$(P)+"[EOF(0)="+STR$(EOF(0))+"]" 
BARCODE 10,Y+25,"39",40,1,0,2,4,"PRICE*"+STR$(Q)+"="+STR$(P*Q) 
Y=Y+100 
PRINT 1 
Y=100 
IF EOF(0)=1 THEN GOSUB OpenData 
IF EOF(0)=0 THEN GOTO Start 
END 
:OpenData 
IF I=1 THEN OPEN "DATA1",0 
IF I=2 THEN OPEN "DATA2",0 
SEEK 0,0 
IF I>2 THEN END 
I=I+1 
RETURN 
EOP 
DEMO 
 
储存上述资料内容并命其档名为”DEMO”，下述步骤可将”DEMO”档存于打印机内并执行

该文件。 
 
<于 DOS 模式下>： 
C:\>COPY DEMO/B LPT1  
 
执行打印机内”DEMO.BAS”文件： 
C:\>COPY CON LPT1 
DEMO 
<Ctrl><Z> 
 
上述范例可将”DATA1”及”DATA2”以相同的文件编号 (file handle) 0 开启，并读出文件内

的内容。 
 

其他参考项目： 
DOWNLOAD, EOP, READ, EOF, LOF, SEEK, FREAD$() 

 READ 
 
功能：读出已开启文件的资料内容逐行读出，并存于变量之中。 

 
语法： 

READ file handle, variables 
 
参     数   

 

说       明 
file handle   0 或 1 
variables   字符串变量，整数变量或浮点变量 

范例： 
DOWNLOAD "DATA1",20,COMPUTER 
2000 
12 
 
DOWNLOAD "DATA2",16,Mouse 
900 
93 
 
DOWNLOAD "DEMO.BAS" 
SIZE 3,1 
GAP 0,0 
DENSITY 8 
SPEED 4 
DIRECTION 0 
REFERENCE 0,0 
SET PEEL OFF 
I=0 
Y=100 
OPEN "DATA1",0 
OPEN "DATA2",1 
SEEK 0,0 
SEEK 1,0 
:Start 
CLS 
TEXT 10,10,"3",0,1,1,"*****READ COMMAND TEST*****" 
TEXT 10,50,"3",0,1,1,"OPEN-READ DATA"+STR$(I+1) 
ITEM$="" 
READ I,ITEM$,P,Q 
TEXT 10,Y,"2",0,1,1,ITEM$+"$"+STR$(P) 
BARCODE 10,Y+25,"39",40,1,0,2,4,"PRICE*"+STR$(Q)+"="+STR$(P*Q) 
Y=Y+100 
PRINT 1 
Y=100 
IF I<=1 THEN 
  IF EOF(I)=1 THEN 
    I=I+1 
    GOTO Start 
  ELSE 
    GOTO Start 
  ENDIF 
ELSE 
  END 
ENDIF 

EOP 
DEMO 

 
其他参考项目： 

DOWNLOAD, EOP, OPEN, EOF, LOF, SEEK, FREAD$() 

 SEEK 
 

功能：用来控制某已开启文件内的指针位置。 
 

语法： 
SEEK file handle, offset 
 
参     数   

范例： 

说       明 
file handle   0 or 1 
offset   指标位置的偏移量 

 

DOWNLOAD "DATA",12,1234567890 
 
DOWNLOAD "TEST.BAS" 
SIZE 3,1 
GAP 0,0 
DENSITY 8 
SPEED 3 
DIRECTION 1 
REFERENCE 0,0 
CLS 
OPEN "DATA",0 
SEEK 0,4 
READ 0,Num$ 
   TEXT 100,10,"3",0,1,1,"SEEK COMMAND TEST" 
BAR 100,40,300,4 
   TEXT 100,60,"3",0,1,1,"SHIFT 4 CHARACTERS" 
   TEXT 100,110,"3",0,1,1,Num$ 
BAR 100,140,300,4 
SEEK 0,0 
READ 0,Num$ 
   TEXT 100,160,"3",0,1,1,"SHIFT 0 CHARACTERS" 
   TEXT 100,210,"3",0,1,1,Num$ 
PRINT 1 
EOP 
TEST 
 

其他参考项目： 
DOWNLOAD, EOP, OPEN, READ, EOF, LOF, FREAD$() 

 LOF( ) 
 

功能：回传指定文件的大小，以 byte 表示。 
 

语法： 
LOF (“FILENAME”) 
 
参     数   

范例： 

说       明 
FILENAME  已下载至打印机的文件名称 

 

DOWNLOAD "DATA1",10,1234567890 
 

DOWNLOAD "DATA2",15,ABCDEFGHIJKLMNO 
 

DOWNLOAD "LofTest.BAS" 
SIZE 3,3 
GAP 0.08,0 
DENSITY 8 
SPEED 3 
DIRECTION 0 
REFERENCE 0,0 
SET PEEL OFF 
CLS 
  OPEN "DATA1",0 
  OPEN "DATA2",1 
  TEXT 10,20,"4",0,1,1,"LOF() FUNCTION TEST" 
  J=LOF("DATA1") 
  K=LOF("DATA2") 

    
TEXT 10,140,"3",0,1,1,"DATA1 IS: "+STR$(J)+" Bytes" 
TEXT 10,200,"3",0,1,1,"DATA2 IS: "+STR$(K)+" Bytes" 

  PRINT 1 
EOP 

 LofTest 
 
其他参考项目： 

DOWNLOAD, EOP, OPEN, READ, EOF, SEEK, FREAD$() 

 FREAD$( ) 
 

功能：读出一文件内指定 byte 长度的资料。 
 

语法： 
FREAD$ (file handle, byte) 
 
参     数   

范例： 

说          明 
file handle   0 或 1 
byte     指定资料长度，以 byte 计 
 

DOWNLOAD "DATA1",10,1234567890 
 
DOWNLOAD "DATA2",15,ABCDEFGHIJKLMNO 
 
DOWNLOAD "OPEN2.BAS" 
SIZE 3,3 
GAP 0.08,0 
DENSITY 8 
SPEED 3 
DIRECTION 0 
REFERENCE 0,0 
SET PEEL OFF 
CLS 
  Y$="" 
  Z$="" 
  OPEN "DATA1",0 
  OPEN "DATA2",1 
  SEEK 0,0 
  SEEK 1,0 
  TEXT 10,260,"3",0,1,1,"FREAD$(0,6) IS: "+Y$ 
  TEXT 10,320,"3",0,1,1,"FREAD$(1,6) IS: "+Z$ 
  PRINT 1 
EOP 
 

其他参考项目： 
DOWNLOAD, EOP, OPEN, READ, EOF, LOF(), SEEK 

  



 

65 
THERMAL BARCODE PRINTER SERIES 

 

 FOR…NEXT LOOP 
 

功能：重复执行单一行或多行指令，必须定义重复执行的次数，于 FOR…NEXT 执行中途跳离

或者从循环外跳入循环内是不允许的，否则将发生无法预期的错误。 
 

语法： 
For variable = start TO end STEP increment 

statement; start < end 
NEXT 
 
参     数    

范例： 

说          明 
variable    变量名称，最多 8 个字符 
start     定义循环起始值，可为整数或浮点数 
end     定义循环结束值，可为整数或浮点数 
increment        定义循环间隔值，可为整数或浮点数 
 

DOWNLOAD "LOOP.BAS" 
SIZE 3,3 
GAP 0.08,0 
DENSITY 8 
SPEED 4 
DIRECTION 1 
CLS 
FOR I=1 TO 10 STEP 1 
TEXT 100,10+30*(I-1),"3",0,1,1,STR$(I) 
NEXT 
FOR I=1 TO 1000 STEP 100 
TEXT 200,10+((I-1)/10)*3,"3",0,1,1,STR$(I) 
NEXT 
FOR I=110 TO 10 STEP -10 
TEXT 300,10+(ABS(I-110))*3,"3",0,1,1,STR$(I) 
NEXT 
FOR I=1 TO 5 STEP 0.5 
IF I-INT(I)=0 THEN Y=10+60*(I-1) ELSE Y=Y+30 
TEXT 400,Y,"3",0,1,1,STR$(I) 
NEXT 
PRINT 1 
EOP 
LOOP 
 

其他参考项目： 
DOWNLOAD, EOP 

 IF…THEN…ELSE…ENDIF 
 

功能：使用 IF…THEN 判断并执行区段内单一行或多行指令。 
 

语法： 
     单行语法： 

IF condition THEN statement 
 

  注意事项：单一行语法不需使用ENDIF做结尾 
 
或区段语法一： 
 
IF condition THEN 

Statements 
ENDIF 
 
或区段语法二： 
 
IF condition THEN 

Statements 
ELSE 

Statements 
ENDIF 
 
或区段语法三： 
 
IF condition 1 THEN 

Statement block 1 
ELSEIF condition 2 THEN 
  Statement block 2 
 ．．． 
ELSEIF condition n THEN 
  Statement block n 
ENDIF 
 
单行语法 IF…THEN…ELSE 中指令的部分必须小于 255 个字符 
 
参     数   

范例： 

说          明 
condition   判断式，其中可用的关系式为：<, >, =, <=, >= 
statement   语句，每行仅可有一句语句 
 

DOWNLOAD "DEMO.BAS" 
SIZE 3,3 
GAP 0.12,0 
SPEED 4 
DENSITY 8 
DIRECTION 1 
REFERENCE 0,0 
OFFSET 0.00 
SET PEEL OFF 
CLS 
 
A=0 
B=0 

C=0 
D=0 
E=0 
F=0 
G=0 
H=0 
J=0 
K=0 
L=0 
 
FOR I=1 TO 100 
  IF I-INT(I/1)*1=0 THEN A=A+I 
  IF I-INT(I/2)*2=1 THEN B=B+I ELSE C=C+I 
  IF I-INT(I/3)*3=0 THEN 
    D=D+I 
  ENDIF 
  IF I-INT(I/5)*5=0 THEN 
    E=E+I 
  ELSE 
    F=F+I 
  ENDIF 
  IF I-INT(I/7)*7=0 THEN 
    G=G+I 
  ELSEIF I-INT(I/17)*17=0 THEN 
    H=H+I 
  ELSEIF I-INT(I/27)*27=0 THEN 
    J=J+I 
  ELSEIF I-INT(I/37)*37=0 THEN 
    K=K+I 
  ELSE 
    L=L+I 
  ENDIF 
NEXT 
TEXT 100,110,"3",0,1,1,"(1) 1+2+3+...+100="+STR$(A) 
TEXT 100,160,"3",0,1,1,"(2) 1+3+5+...+99="+STR$(B) 
TEXT 100,210,"3",0,1,1,"(3) 2+4+6+...+100="+STR$(C) 
TEXT 100,260,"3",0,1,1,"(4) 3+6+9+...+99="+STR$(D) 
TEXT 100,310,"3",0,1,1,"(5) 5+10+15+...+100="+STR$(E) 
TEXT 100,360,"3",0,1,1,"    (1)-(5)="+STR$(F) 
TEXT 100,410,"3",0,1,1,"(6) 7+14+21+...+98="+STR$(G) 
TEXT 100,460,"3",0,1,1,"(7) 17+34+51+...+85="+STR$(H) 
TEXT 100,510,"3",0,1,1,"(8) 27+54+...+81="+STR$(J) 
TEXT 100,560,"3",0,1,1,"(9) 37+74="+STR$(K) 
TEXT 100,610,"3",0,1,1,"    (1)-(6)-(7)-(8)-(9)="+STR$(L) 
PRINT 1,1 
EOP 
 
DOWNLOAD “IFTHEN.BAS” 
SIZE 4,4 
GAP 0,0 
DENSITY 8 
SPEED 3 
DIRECTION 0 
REFERENCE 0,0 
SET PEEL OFF 
CLS 

A=50 
B=5 
C$="" 
D$="" 
 
:L1 
IF A>100 THEN GOTO L1 ELSE A=A+10 
C$=STR$(A)+" IS SMALLER THAN 100" 
TEXT 100,10,"4",0,1,1,C$ 
PRINT 1 
END 
 
:L2 
A=A+B 
D$=STR$(A)+" IS LARGER THAN 100" 
TEXT 100,100,"4",0,1,1,D$ 
PRINT 1 
GOTO L1 
EOP 
 
注意事项： 
单行语法：当IF后的判断式为真时，执行THEN后方的语句；为假时，执行ELSE后方的语

句。若无ELSE，则程序将执行IF…THEN之后的语句。 
区段语法：必须以ENDIF结尾 
 

其他参考项目： 
DOWNLOAD, EOP 

 GOSUB…RETURN 
 

功能：跳至某一标签(子程序)执行，执行完成后遇到”RETURN”时再跳回原位置继续执行下一

行程序。 
 

语法： 
GOSUB LABEL 

statement 
END 
:LABEL 

statement 
RETURN 
 
参     数   

 

说        明 
LABEL   子程序的起始点，标签名称最长 8 个字符 

范例： 

DOWNLOAD "GOSUB1.BAS" 

SIZE 3,3 

GAP 0,0 

DENSITY 8 

SPEED 4 

DIRECTION 0 

CLS 

TEXT 10,10,"3",0,1,1,"GOSUB & RETURN COMMAND TEST" 

GOSUB DR_BOX 

PRINT 1 

END 

:DR_BOX 

  FOR I=21 TO 81 STEP 10 

      BOX 80+I,80+I,80+300-I,80+300-I,5 

  NEXT 

RETURN 

EOP 

GOSUB1 

 
其他参考项目： 

DOWNLOAD, EOP, END, GOTO 

 GOTO 
 

功能：跳至某一标签位置往下执行，标签名称最长 8 个字符 
 

语法： 
GOTO LABEL 
 
:LABEL 
statement 
 

范例： 
DOWNLOAD "GOTO1.BAS" 
SIZE 3,3 
GAP 0.08,0 
DENSITY 8 
SPEED 4 
DIRECTION 1 
REFERENCE 0,0 
SET PEEL OFF 
CLS 
A=0 
TOTAL=0 
:START 
  IF A<100 THEN  
    GOTO SUM  
  ELSE  
    GOTO PRTOUT 
  ENDIF 
:SUM 
  A=A+1 
  TOTAL=TOTAL+A 
  GOTO START 
:PRTOUT 
  B$="THE SUMMATION OF 1..100 IS "+STR$(TOTAL) 
  TEXT 10,100,"3",0,1,1,B$ 
  PRINT 1 
END 
EOP 
 

其他参考项目： 
DOWNLOAD, EOP, END, GOSUB…RETURN 

 REM 
 
    功能：注解用，于字首加注”REM”则该行将被忽略不执行。 
 

语法： 
REM string 
 

范例： 
REM ******************************** 
REM This is a demonstration program* 
REM ********************************  
DOWNLOAD "REMARK.BAS" 
SIZE 3,3 
GAP 0.08,0 
DENSITY 8 
SPEED 4 
DIRECTION 1 
REFERENCE 0,0 
SET PEEL OFF 
CLS 
TEXT 50,50,"3",0,1,1,"REMARK DEMO PROGRAM" 
REM TEXT 50,100,"3",0,1,1,"REMARK DEMO PROGRAM" 
PRINT 1,1 
EOP 
 

其他参考项目： 
DOWNLOAD, EOP, END 

 INT( ) 
 

功能：回传浮点数的整数部分，小数部分则删去。 
 

语法： 
INT (n) 
 
参     数   

范例： 

说          明 
n    n 可以是正整数、负整数、浮点数或数学运算式  
 

DOWNLOAD "DEMO.BAS" 
SIZE 4,2 
GAP 0,0 
DENSITY 8 
SPEED 3 
DIRECTION 0 
REFERENCE 0,0 
SET PEEL OFF 
CLS 
REM **** To round up or down**** 
INPUT "Number:",Num 
N=INT(Num+0.5) 
IF N>Num THEN 
  TEXT 50,100,"3",0,1,1,"To round up= "+STR$(N) 
ELSE 
  TEXT 50,100,"3",0,1,1,"To round down= "+STR$(N) 
ENDIF 
PRINT 1 
EOP 

 
其他参考项目： 

DOWNLOAD, EOP, END, ABS(), ASC(), STR$() 

 LEFT$( ) 
 

功能：此函数可回传一字符串中由左端向右指定长度的字符串。 
  

 语法： 
LEFT$ (X$, n) 
 
参     数    

范例： 

说        明 
X$     待处理的字符串 
n     自左端向右算起要回传的字符串长度 

 

DOWNLOAD "STR1.BAS" 
SIZE 3.00,3.00 
GAP 0.08,0.00 
SPEED 4.0 
DENSITY 8 
DIRECTION 0 
REFERENCE 0,0 
CLS 
A$="BARCODE PRINTER DEMO PRINTING" 
C$=LEFT$(A$,10) 
TEXT 10,10,"3",0,1,1,A$ 
TEXT 10,100,"3",0,1,1,"10 LEFT 10 CHARS: "+C$ 
PRINT 1 
EOP 
 

其他参考项目： 
DOWNLOAD, EOP, END, RIGHT$(), MID$(), LEN(), STR$() 

 LEN( ) 
 

功能：回传一字符串的长度 
语法： 

LEN (string) 
 
参     数   

 

说          明 
string    待回传长度的字符串 

范例： 
DOWNLOAD "DEMO.BAS" 
SIZE 3.00,3.00 
GAP 0.08,0.00 
SPEED 4.0 
DENSITY 8 
DIRECTION 0 
REFERENCE 0,0 
CLS 
A$="TAIWAN SEMICONDUCTOR CO., LTD" 
B=LEN(A$) 
TEXT 10,10,"3",0,1,1,A$ 
TEXT 10,50,"3",0,1,1,"STRING LENGTH="+STR$(B) 
PRINT 1 
EOP 
 

其他参考项目： 
DOWNLOAD, EOP, END, LEFT$(), LEN(), RIGHT$(), MID$(), STR$(), VAL() 

 MID$( ) 

 语法： 
MID$(string, m, n) 
 
参     数   

 范例： 

说          明 
string    待处理的字符串 
m    自字符串的第 m 个字符开始选取，其中 1≦m≦字符串长度 
n    回传的字符串长度 
 

DOWNLOAD "DEMO.BAS" 
SIZE 3.00,3.00 
GAP 0.08,0.00 
SPEED 4.0 
DENSITY 8 
DIRECTION 0 
REFERENCE 0,0 
 
CLS 
A$="TAIWAN SEMICONDUCTOR CO., LTD" 
E$=MID$(A$,11,10) 
TEXT 10,10,"3",0,1,1,A$ 
TEXT 10,200,"3",0,1,1,"10 MIDDLE CHARS: "+E$ 
PRINT 1 
EOP 
 

其他参考项目： 
DOWNLOAD, EOP, END, LEFT$(), LEN(), RIGHT$(), STR$(), VAL() 

 RIGHT$( ) 
 

功能：此函数可回传一字符串中由右端向左指定长度的字符串。 
  
 语法： 

RIGHT$ (X$, n) 
 
参     数   

 范例： 

说       明 
X$    待处理的字符串 

n     自右端向左算起要回传的字符串长度 
 

DOWNLOAD "DEMO.BAS" 
SIZE 3.00,3.00 
GAP 0.08,0.00 
SPEED 4.0 
DENSITY 8 
DIRECTION 0 
REFERENCE 0,0 
 
CLS 
A$="TAIWAN SEMICONDUCTOR CO., LTD" 
D$=RIGHT$(A$,10) 
TEXT 10,10,"3",0,1,1,A$ 
TEXT 10,150,"3",0,1,1,"10 RIGHT CHARS: "+D$ 
PRINT 1 
EOP 
 

其他参考项目： 
DOWNLOAD, EOP, END, LEFT$(), LEN(), MID$(), STR$(), VAL() 

 STR$( ) 
 

功能：将某一数值转换为字符串格式 
 

 语法： 
STR$ (n) 
 
参     数   

 范例： 

说       明 
n    可为一整数、浮点数或数学运算式 

  

DOWNLOAD "DEMO.BAS" 
SIZE 3.00,3.00 
GAP 0,0.00 
SPEED 4.0 
DENSITY 8 
DIRECTION 0 
REFERENCE 0,0 
CLS 
A$="TAIWAN SEMICONDUCTOR CO., LTD" 
F=100 
G=500 
H$=STR$(F+G) 
TEXT 10,10,"3",0,1,1,A$ 
TEXT 10,60,"3",0,1,1,"F="+STR$(F) 
TEXT 10,110,"3",0,1,1,"G="+STR$(G) 
TEXT 10,160,"3",0,1,1,"F+G="+H$ 
PRINT 1 
EOP 
DEMO 

 
 

其他参考项目： 
DOWNLOAD, EOP, END, LEFT$(), LEN(), RIGHT$(), MID$(), VAL() 

 VAL( ) 
 

功能：数值字符串转换为整数型态或浮点数型态 
 

 语法： 
VAL (“numeric character”) 
 
参     数    

范例： 

说          明 
numeric character  数字 0~9 及小数点“.” 

 

DOWNLOAD "DEMO.BAS" 
SIZE 3.00,3.00 
GAP 0.00,0.00 
SPEED 4.0 
DENSITY 8 
DIRECTION 0 
REFERENCE 0,0 
CLS 
 A$="TAIWAN SEMICONDUCTOR CO., LTD" 
 F$="100" 
 G$="500" 
 H=VAL(F$)+VAL(G$) 
 I$=STR$(H) 
 TEXT 10,10,"3",0,1,1,A$ 
 TEXT 10,60,"3",0,1,1,"F="+F$ 
 TEXT 10,110,"3",0,1,1,"G="+G$ 
 TEXT 10,160,"3",0,1,1,"F+G="+I$ 
 PRINT 1 
EOP 
DEMO 

 
 

其他参考项目： 
DOWNLOAD, EOP, END, LEFT$(), LEN(), RIGHT$(), MID$(), STR$() 

 BEEP 
 

功能：打印机收到该指令会响一声 
 

语法： 
BEEP 
 

范例： 
 
DOWNLOAD “DEMO.BAS” 
SIZE 4,4 
GAP 0,0 
DENSITY 8 
SPEED 6 
DIRECTION 0 
REFERENCE 0,0 
SET PEEL OFF 
CLS 
BEEP 
INPUT “Text1 =”,TEXT1$ 
TEXT 100,100,”3”,0,1,1,TEXT1$ 
PRINT 1 
EOP 

打印机外围功能设定指令 
 
 SET COUNTER  
 

功能：COUNTER 可以是一般的计数器或是一变量。可设定计数器的初始值及计数使用的增量。 
 

 语法： 
SET COUNTER @n step  
@n = “Expression” 
 
参     数    

 范例： 

说          明 
@ n: 计数器的初始值。总计有 50(@0~@49)个计数器可同时使

用  
step     计数器的增量值可为正数亦可为负数。 

-999999999<= step <=999999999 
如果计数器只是被当做固定的变量使用时，请将增量值设为 0 

Expression   初始字符串 
 

SIZE 3,3 
GAP 0,0 
DENSITY 8 
SPEED 6 
DIRECTION 0 
REFERENCE 0,0 
SET COUNTER @1 1 
@1="00001" 
SET COUNTER @2 5 
@2="AB000001" 
CLS 
TEXT 50,50,"3",0,1,1,@1 
BARCODE 50,100,"39",48,1,0,2,4,@2 
PRINT 2,1 
 

其他参考项目： 
PRINT, TEXT, BARCODE 

 SET KEY1, SET KEY2 
 

功能：设定开启/关闭 KEY1/KEY2 的功能。 
  

 语法： 
SET KEY1 ON/OFF 
SET KEY2 ON/OFF 
 
参     数    

 范例： 

说          明 
ON     开启 KEY1 定义为暂停功能 
      开启 KEY2 定义为进纸功能 
OFF     关闭 KEY1 定义为暂停功能 
      关闭 KEY2 定义为进纸功能 
 
注意事项：设定值将会被记录在打印机中 
 

DOWNLOAD "DEMO.BAS" 
SIZE 3,1 
GAP 0,0 
DENSITY 8 
SPEED 3 
DIRECTION 0 
REFERENCE 0,0 
SET KEY1 OFF 
SET KEY2 OFF 
KEY1=0 
KEY2=0 
 
CLS 
:START 
 
IF KEY1=1 THEN 
  CLS 
  TEXT 100,10,"3",0,1,1,"KEY1 (PAUSE key) is pressed!!" 
  PRINT 1,1 
ELSEIF KEY2=1 THEN  
  CLS 
  TEXT 100,10,"3",0,1,1,"KEY2 (FEED key) is pressed!!" 
  PRINT 1,1 
  SET KEY1 ON 
  SET KEY2 ON 
  END 
ENDIF 
GOTO START 
EOP 
DEMO 
 

其他参考项目： 
OFFEST, PRINT 

 SET PEEL 
 

功能：设定启动/关闭自动剥纸器功能。预设值为关闭状态，当此功能被开启时，打印机会在每

印完一张时即暂停，直到标签纸被取走后才会打印下一张标签。此项设定会被记录于打

印机中。 
 

 语法： 
SET PEEL ON/OFF 
 
参     数    

 范例： 

说          明 
ON     开启自动剥纸器的功能 
OFF     关闭自动剥纸器的功能 
 

REM ***SELF-PEELING FUNCTION ON*** 
SIZE 4,4 
GAP 0,0 
DENSITY 8 
SPEED 6 
DIRECTION 0 
REFERENCE 0,0 
SET PEEL ON 
CLS 
TEXT 50,100,"3",0,1,1,"SELF-PEELING FUNCTION TEST" 
PRINT 5 
 

其他参考项目： 
OFFEST, PRINT 

 SET TEAR & SET STRIPPER 
 

功能：设定开启/关闭送纸至撕纸线的功能，此项设定会被记录于打印机中 
 

 语法： 
SET TEAR ON/OFF 
 
参     数    

 范例： 

说       明 
ON 标签打印结束时将送纸至撕纸位置 
OFF 标签打印结束时会将标签起印点停留至打印线位置 
 

REM ***TEAR FUNCTION ON*** 
SIZE 3,3 
GAP 0.08,0 
DENSITY 8 
SPEED 4 
DIRECTION 0 
REFERENCE 0,0 
SET PEEL OFF 
SET TEAR ON 
CLS 
TEXT 50,100,"3",0,1,1,"TEAR FUNCTION TEST" 
PRINT 1 
 

其他参考项目： 
SET PEEL 

 SET HEAD 
 

功能：设定开启/关闭打印头开启感应器。若感应器为关闭状态，则当打印头被开启时将不会有

任何错误讯息出现，且该设定会被存于打印机中。 
 

 语法： 
SET HEAD ON /OFF 
 
参     数    

 范例： 

说       明 
ON     开启打印头感应器 
OFF     关闭打印头感应器 
 

SET HEAD ON 
SET HEAD OFF 

 SET RIBBON 
 

功能：设定开启/关闭碳带感应器，即切换热转式/热感印式打印。通常打印机于开启电源时，

碳带感应器即会自动检测打印机是否已装上碳带，并藉此决定使用热感式或热转式打印。

此项设定并不会存于打印机中。此方法仅适用于热转式机器。 
 

 语法： 
SET RIBBON ON /OFF 
 
参     数    

 范例： 

说      明 
ON     热转式打印 
OFF     热感式打印 
 

REM ***Direct printing**** 
SIZE 4,4 
GAP 0,0 
DENSITY 8 
SPEED 6 
DIRECTION 0 
REFERENCE 0,0 
SET PEEL OFF 
SET RIBBON OFF 
CLS 
BARCODE 100,100,"39",48,1,0,2,5,"CODE 39” 
PRINT 1 

 SET COM1 
 

功能：设定打印机串口的通讯参数。 
 

 语法： 
SET COM1 baud, parity, data, stop 
 
参     数   

 范例： 

说          明 
baud    波特率，可用的波特率速度如下： 

24: 2400 bps 
48: 4800 bps 
96: 9600 bps 
19: 19200 bps 

 
parity    校验位 
      N: 无 
      E: 偶数 
      O: 奇数 
 
data    数据位 
      8: 8 位资料 
      7: 7 位资料 
 
stop    停止位 
      1 
      2 
 

通常于 DOS 下透过并行口(Parallel port)设定打印机的串口通讯参数。 
 
C:\>COPY CON LPT1<ENTER> 
SET COM1 19,N,8,1<ENTER> 
<CTRL><Z><ENTER> 
C:\> 
 

注意事项：上方范例中<ENTER>表示按下键盘上的”ENTER”键，<CTRL>表示按下
键盘上的”Ctrl”键，<Z>表示按下键盘上的”Z”键。 

 SET PRINTKEY 
 

功能：此设定会打印一张标签，并将标签前推至撕纸位置。当按下 FEED 键时会打印下一张标

签内容或是下一批某数量的标签。若标签内容含有变化的文字或条码时，序号的部分将

会依序改变，此项设定将被记录于打印机中。 
 

 语法： 
 SET PRINTEKY OFF/ON/AUTO/<num> 
 
参     数   
OFF    关闭此项功能 

说          明 

ON    开启此项功能 
AUTO   自动开启此项功能 
<num>   设定当按下 FEED 键时会打印的张数 
 

范例： 
先执行下方程序： 
SIZE 4,2.5 
GAP 0.12,0 
SET FEEDPRINT ON 
SET COUNTER @0 1 
@0="0001" 
CLS 
TEXT 10,10,"5",0,1,1,@0 
PRINT 1 

 
再执行下列动作： 

语     法 打印机收到指令 “PRINT m” 打印结果 
SET PRINTKEY ON or 
SET PRINTKEY AUTO 

1.) PRINT 2 Label 1~2 
2.) 按下 FEED 键 Label 3~4 

 
语     法 打印机收到指令 “PRINT m,n” 打印结果 

SET PRINTKEY ON or 
SET PRINTKEY AUTO 

1.) PRINT 1,2 Label 1, Label 1 
2.) 按下 FEED 键 Label 2, Label 2 

 
语    法 打印机收到指令 “PRINT -1,n” 打印结果 

SET PRINTKEY ON or 
SET PRINTKEY AUTO 

1.) PRINT -1,2 Label 1, Label 1 
2.) 按下 FEED 键 Label 1, Label 1 

        
语    法 打印机收到指令 “PRINT m” 打印结果 

SET PRINTKEY 5 
1.) PRINT 2 Label 1~2 
2.) 按下 FEED 键 Label 3~7 

 
 

语    法 打印机收到指令 “PRINT m,n” 打印结果 

SET PRINTKEY 5 
1.) PRINT 1,2 Label 1, Label 1 
2.) 按下 FEED 键 Label 2~6 

 
语    法 打印机收到指令 “PRINT –1,n” 打印结果 

SET PRINTKEY 5 
1.) PRINT -1,2 Label 1, Label 1 
2.) Press FEED key Label 1, Label 1 

 SET REPRINT 
 

功能：设定当发生”缺纸”、”卡纸”、”缺碳带”或”打印头开启”等错误时，于回复正常状态时是否

要重印上一张未打印完成的资料。 
 

 语法： 
SET REPRINT OFF/ON 
 
参     数   
OFF    禁用此功能 

说      明 

ON    启动此功能 
 

范例： 
SET REPRINT ON 

 PEEL  
 

功能：获得剥纸感应器的状态，状态值仅可读。 
 

 语法： 
      PEEL 
 

回  传  值    

 范例： 

说          明 
0                     剥纸感应器上方无纸 
1      剥纸感应器上方有纸 
 

DOWNLOAD "DEMO.BAS" 
SIZE 4,1 
GAP 0,0 
SPEED 4 
DENSITY 8 
SET PEEL OFF 
SET KEY1 OFF 
:START 
  IF KEY1=1 THEN GOTO A 
GOTO START 
:A 
CLS 
TEXT 10,10,"3",0,1,1,"PEEL Function Test!!" 
PRINT 1,1 
 
:B 
IF PEEL=1 THEN  
  GOTO B  
ELSE 
  CLS 
  TEXT 10,10,"3",0,1,1,"The label is removed from the PEEL sensor!!" 
  PRINT 1,1 
  GOTO START 
ENDIF 
EOP 
DEMO 

 KEY1, KEY2 
 

功能：用来取得按键状态的值。 
 
 语法： 
  KEYm=n 
 

按     键     

 范例： 

回     传     值 
KEY1 (PAUSE)    0: 未按住 

1: 按  住 
KEY2 (FEED)     0: 未按住 

1: 按  住 
 

DOWNLOAD "DEMO.BAS" 
SIZE 3,1 
GAP 0,0 
SPEED 4 
DENSITY 8 
DIRECTION 1 
REFERENCE 0,0 
SET KEY1 OFF 
:START 
IF KEY1=1 THEN 
CLS 
 TEXT 100,10,"3",0,1,1,"KEY FUNCTION TEST" 
 PRINT 1,1 
ELSE 
ENDIF 
GOTO START 
EOP 
DEMO   

打印机全域变量 
 
 @LABEL 
 

功能：用来计算打印机总共打印的张数。会记录于打印机中，且关开机时不会归零。 
 

 语法： 
写入格式： @LABEL=n 或 @LABEL=”n” 
读出格式：A=LABEL 或 A$=STR$(LABEL) 
 
参     数    

 范例： 

说          明 
n      已打印的标签数量，其中 0<=n<=999999999 
 

DOWNLOAD "DEMO.BAS" 
SIZE 4,2.5 
GAP 2 mm,0 
SPEED 6 
DENSITY 12 
CLS 
TEXT 10,50,"3",0,1,1,@LABEL 
TEXT 10,100,"3",0,1,1,"@LABEL="+STR$(LABEL) 
TEXT 10,150,"3",0,1,1,"*****Statement 1*****" 
  IF LABEL>1000 THEN 
    TEXT 10,200,"3",0,1,1,"LABEL>1000" 
  ELSE 
    TEXT 10,200,"3",0,1,1,"LABEL<1000" 
  ENDIF 
TEXT 10,250,"3",0,1,1,"*****Statement 1*****" 
  A=LABEL 
  IF A>1000 THEN 
     TEXT 10,300,"3",0,1,1,"A>1000" 
  ELSE 

TEXT 10,300,"3",0,1,1,"A<1000" 
  ENDIF 
TEXT 10,350,"3",0,1,1,"*****Statement 3*****" 
  A$=STR$(LABEL) 
  IF VAL(A$)>1000 THEN 
    TEXT 10,400,"3",0,1,1,"VAL(A$)>1000" 
  ELSE 
    TEXT 10,400,"3",0,1,1,"VAL(A$)<1000" 
  ENDIF 
PRINT 1,1 
EOP 

